Maple Basics for Calculus I

Maple is a general purpose mathematics software package. When you start the program, it produces a screen and types a prompt (usually `>`) which is Maple's way of telling you that it is ready to accept a command.

To start Maple in the campus labs, find it in the "start menu" under Mathematics. (If it is not there, try navigating to H:Maple51/bin.wnt/ and click on wmaple.exe). You probably will want to start by "maximizing" the working window inside of the main Maple window that comes up.

Once Maple has started, you type a command, *followed by a semicolon*, and then press the return or enter key. It seems clumsy at first to have to put a semicolon after each command. But this allows you to enter a more complex command on several lines, or to put several commands in succession on a single line. In most versions of maple, you can edit and re-execute previous commands by using a mouse or the arrow keys.

Expressions are specified in "computerese", `+` and `-` have their usual meaning, `*` means multiply, and `/` means divide. You'll often need to use lots of parenthesis to make sure the order of operation is correct, but `*` and `/` bind more tightly than `+` and `-`. *Don't* use square brackets `[ ]` for grouping, because Maple reserves these brackets for vectors. A variety of sample commands appear below to illustrate maple usage.

`factor(x^12-1);` factors $x^{12}-1$

`expand( (x^7-5)*(x^2+3*x-5)*(2*x-9) );` multiplies out $(x^7-5)(x^2+3x-5)(2x-9)$

`solve(x^2+5*x-17*a=0, x);` solves $x^2+5x-17a=0$ for x

`fsolve(x^6+x-1=0, x);` finds approximate (floating point) solutions to $x^6+x-1=0$

`limit( (x^2-1)/(x-1), x=1);` computes $\lim_{x\to 1} \frac{x^2-1}{x-1}$

`limit(abs(x)/x, x=0, right);` computes $\lim_{x\to 0^+} \frac{|x|}{x}$

`diff(x^2+1, x);` computes $\frac{d}{dx}\left(x^2+1\right)$

`diff(sin(sqrt(x))/cos(2*Pi*x), x);` computes $\frac{d}{dx}\left(\frac{\sin(\sqrt{x})}{\cos(2\pi x)}\right)$

`evalf(Pi);` approximates π

`evalf(Pi,30);` approximates π to 30 decimal places

`plot(x^2-1, x=-2..3);` plots $y=x^2-1$ for $x\in[-2,3]$

`plot({x, x^2, x^3}, x=-1..1);` plots x, x^2, and x^3 together on the same set of axes

`plot(x^2-1, x=-2..3, y=0..5);` view of previous plot with y restricted to $[0,5]$

Spaces are usually ignored, you can put them in to make it easier to read what you are typing. The `%` symbol is a shorthand for the result of the last expression computed. (Versions of Maple before V.5 used `"` instead.) For example, use `evalf(%)` to get a decimal approximation of the last result. Other common uses are `factor(%)`, `expand(%)`, and especially `simplify(%)`. (*Warning:* the `%` refers to the last command which Maple has executed. If you use a mouse or the arrow keys to move around and re-execute commands, this may not be the command which appears directly above the current line!)

The command `y:=x^2-1;` assigns the *expression* x^2-1 to the *variable* y. This can be useful if you want to do several things to an expression. For example, `diff(y,x);` would differentiate x^2-1, and `plot(y, x=-1..2);` would then graph it. Variable names are not limited to a single letter. (Note the distinction between `=`, which asserts that two things should be equal, and `:=` which assigns the value of the thing on the right to the left hand one. Mathematicians use $=$ for both of these meanings, relying on the context to make clear what is meant, but computers are not quite that smart yet.)

The expression `f:= x -> x^2-1;` defines the *function* $f(x)=x^2-1$. This differs from the above example in that you can now compute `f(2)`, `f(3)`, and even `f(x+2)`. Note that `f(x):=x^2-1` will not work correctly. Instead you need to assign to the variable f the expression `x -> x^2-1`, which is maplese for the function that sends a number x to the value x^2-1.

Maple also has an extensive help system built in. You can use the help menu to get help screens which are organized by topic. Or, you can use the `?` command. For example, `?plot` will bring up the help screen on the plot command.

(over)

Exercises

1) To get warmed up, spend about 5 minutes trying out at least some of the sample commands on the "Maple Basics" handout.

2) Run the following sequence of commands and explain what happens.

```
f:= x -> x^12;
(f(x+h)-f(x))/h;
simplify(%);
limit(%, h=0);
```

3) Use Maple to work compute some of the derivative exercises on pages 175–176 of the textbook. Use Maple to plot some of the graphs from pages 230–231.

4) Use maple to plot the graph of $f(x) = \sqrt{x} + 1/x^2$. Then run the commands `f:= sqrt(x) + 1/x^2;`, `fp:= diff(f, x);`, `solve(fp=0,x);`, and `fsolve(fp=0,x);`. What does the last answer correspond to on the graph? (Note that Maple uses the symbol `I` for the "imaginary" square root of -1.) Can you find the y-value of the minimum point for f?

5) Use Maple to find the equation of the line tangent to the graph of $y = x^3 - x$ at $x = 2$. Can you get Maple to do all the work—finding the slope and y-intercept? Sketch the curve and tangent line on a single set of axes.

6) Maple has a built in command for defining piecewise functions. Try `y:= piecewise(x<=0, x^2, x>0, 2*x);` and then `plot(y, x=-2..2);` Note that each condition is placed **before** the equation it applies to, not after! Also, Maple uses `<=` and `>=` for $\leq$ and $\geq$. Maple will not accept "double inequalities" such as `0<=x<=1`, you need to use `(0<=x and x<=1)` instead. (Maple recognizes the word `and` as a "Boolean operator.") Use `piecewise` to define and plot the function for the height of the ball in Quiz 7. Then use `diff` to compute and plot the velocity and acceleration. Does Maple get the correct answer?

7) Use Maple to sketch graphs of $\sin(1/x)$, $x\sin(1/x)$, $x^2\sin(1/x)$, and $x^3\sin(1/x)$. Use Maple to differentiate each of these, and to plot the derivatives. You may need to increase the number of points Maple will use in order to get good graphs. Then command `plot(sin(1/x), x=0 .. 0.2, resolution=400);` will (perhaps slowly) plot the graph using lots of points—twice the default resolution of 200.

8) Use Maple to solve the extra-credit from Quiz 8. To compute $\lim_{x\to\infty} f(x)$, use `limit(f, x=infinity);`

Here are some more examples which will be useful later this semester and in Calc II and Calc III:

`integrate(5*x^3+2, x);` computes $\int 5x^3 + 2\,dx$ (shorthand form: `int(5*x^3+2, x);`)

`integrate(exp(4*y), y=0..1);` computes $\int_0^1 e^{4y}\,dy$

`seq(i^2, i=1..20);` prints out the values of 1^2, 2^2, 3^2, ..., 20^2

`sum(i^2, i=1..20);` computes $\sum_{i=1}^{20} i^2$

`integrate(1/(x^2+3*x), x=1..infinity);` computes $\int_1^\infty \frac{1}{x^2+3x}\,dx$

`convert(1/(x^2+3*x), parfrac, x);` converts $\frac{1}{x^2+3x}$ to partial fractions form

`sum(1/(i^2+3*i), i=1..infinity);` computes $\sum_{i=1}^{\infty} \frac{1}{i^2+3i}$

`plot3d(x^2+y^2, x=-1..1, y=-1..1);` plots $z = x^2 + y^2$ for $(x, y) \in [-1, 1] \times [-1, 1]$